# $Id: PKGBUILD 138168 2011-09-18 03:14:27Z eric $
# Maintainer: Thayer Williams <thayer@archlinux.org>
# Contributor: Eric Johnson <eric@coding-zone.com>

pkgname=i8kutils
pkgver=1.33
pkgrel=2
pkgdesc="Dell Inspiron/Latitude kernel driver and utilities"
arch=('i686' 'x86_64')
url="http://people.debian.org/~dz/i8k/"
license=('GPL2')
depends=('glibc')
source=(ftp://ftp.archlinux.org/other/${pkgname}/${pkgname}_${pkgver}.tar.gz)
md5sums=('1ab077b301dbb0f4be8ef52eb3b97628')

build() {
  cd "$srcdir/$pkgname-$pkgver"
  make
}

package() {
  cd "$srcdir/$pkgname-$pkgver"
  install -D -m755 i8kctl "$pkgdir/usr/bin/i8kctl"
  install -D -m755 i8kbuttons "$pkgdir/usr/bin/i8kbuttons"
  install -D -m644 i8kbuttons.1 "$pkgdir/usr/share/man/man1/i8kbuttons.1"
  install -D -m644 i8kctl.1 "$pkgdir/usr/share/man/man1/i8kctl.1"
  ln -fs ./i8kctl "$pkgdir/usr/bin/i8kfan"
}
